Using Mathematica's Around[ ] function for Uncertainties

In[7]:=
```
Clear["Global`*"]; SetDirectory[NotebookDirectory[]]; DateString[]
```

Out[7]=
```
Wed 26 Mar 2025 12:13:31
```

Consider this data from the Torsional Oscillator Experiment.

In[8]:=
```
{ω0, δω0} = {4.41458, 3 * 10^(-5)};
{ωv, δωv} = {4.40803, 2 * 10^(-5)};
{γ, δγ} = {0.4359, 1 * 10^(-4)};
```

You can use the Around function to express the number with its uncertainty.

In[13]:=
```
Around[ω0, δω0]
```

Out[13]=
```
4.414580 ± 0.000030
```

Here is the function for computing Q

In[12]:=
```
Q[ω0_, γ_] := ω0 / γ
```

You can also use the 'Around' function to do calculations with uncertainty.

In[14]:=
```
Q[Around[ω0, δω0], Around[γ, δγ]]
```

Out[14]=
```
10.1275 ± 0.0023
```

Or, calculating the uncertainty the long way:

In[16]:=
```
Q[ω0, γ] * Sqrt[(δω0/ω0)^2 + (δγ/γ)^2]
```

Out[16]=
```
0.00232437
```

This also works for messier equations, such as that for finding Q from the difference of the two frequencies as long as the uncertainties are all "small".

In[21]:=
```
Clear[Q1]
Q1[ω0_, ωv_] := ω0 / (2 Sqrt[ω0^2 - ωv^2])
```

In[23]:=
```
Q1[ω0, ωv]
```

Out[23]=
```
9.18206
```

In[24]:=
```
Q1[Around[ω0, δω0], Around[ωv, δωv]]
```

Out[24]=
```
9.182 ± 0.025
```